#version 450 core

struct Material
{
  sampler2D diffuse;
  sampler2D specular;
  float     shininess;
}; 

struct DirLight
{
    vec3 direction;
  
    vec3 ambient;
    vec3 diffuse;
    vec3 specular;
};

struct PointLight
{    
    vec3 position;
    
    float constant;
    float linear;
    float quadratic;  

    vec3 ambient;
    vec3 diffuse;
    vec3 specular;
};

struct SpotLight
{
    vec3 position;
    vec3 direction;
    float cutOff;
    float outerCutOff;
  
    float constant;
    float linear;
    float quadratic;
  
    vec3 ambient;
    vec3 diffuse;
    vec3 specular;       
};

out vec4 color;

uniform vec3 viewPos;
uniform Material material;

#define NR_DIR_LIGHTS 1
#define NR_POINT_LIGHTS 4
uniform DirLight dirLight;
uniform PointLight pointLights[NR_POINT_LIGHTS];
uniform SpotLight spotLight;

in vec3 vNormal;
in vec3 vPos;
in vec2 vTexCoords;

vec3 CalcDirLight(DirLight light, vec3 normal, vec3 viewDir);
vec3 CalcPointLight(PointLight light, vec3 normal, vec3 fragPos, vec3 viewDir);
vec3 CalcSpotLight(SpotLight light, vec3 normal, vec3 fragPos, vec3 viewDir);

void main()
{
  // properties
  vec3 norm = normalize(vNormal);
  vec3 viewDir = normalize(viewPos - vPos);
  
  // == =====================================================
  // Our lighting is set up in 3 phases: directional, point lights and an optional flashlight
  // For each phase, a calculate function is defined that calculates the corresponding color
  // per lamp. In the main() function we take all the calculated colors and sum them up for
  // this fragment's final color.
  // == =====================================================
  // phase 1: directional lighting
  vec3 result = CalcDirLight(dirLight, norm, viewDir);
  // phase 2: point lights
  for(int i = 0; i < NR_POINT_LIGHTS; i++)
    result += CalcPointLight(pointLights[i], norm, vPos, viewDir);    
  // phase 3: spot light
  result += CalcSpotLight(spotLight, norm, vPos, viewDir);    
  
  color = vec4(result, 1.0);
}

vec3 CalcDirLight(DirLight light, vec3 normal, vec3 viewDir)
{
  vec3 lightDir = normalize(-light.direction);
  
  // diffuse shading
  float diff = max(dot(normal, lightDir), 0.0);
  
  // specular shading
  vec3 reflectDir = reflect(-lightDir, normal);
  float spec = pow(max(dot(viewDir, reflectDir), 0.0), material.shininess);
  
  // combine results
  vec3 ambient  = light.ambient  * vec3(texture(material.diffuse, vTexCoords));
  vec3 diffuse  = light.diffuse  * diff * vec3(texture(material.diffuse, vTexCoords));
  vec3 specular = light.specular * spec * vec3(texture(material.specular, vTexCoords));
  return (ambient + diffuse + specular);
}

vec3 CalcPointLight(PointLight light, vec3 normal, vec3 fragPos, vec3 viewDir)
{
  vec3 lightDir = normalize(light.position - fragPos);
  
  // diffuse shading
  float diff = max(dot(normal, lightDir), 0.0);
  
  // specular shading
  vec3 reflectDir = reflect(-lightDir, normal);
  float spec = pow(max(dot(viewDir, reflectDir), 0.0), material.shininess);
  
  // attenuation
  float distance    = length(light.position - fragPos);
  float attenuation = 1.0 / (light.constant + light.linear * distance + light.quadratic * (distance * distance));
  
  // combine results
  vec3 ambient  = light.ambient  * vec3(texture(material.diffuse, vTexCoords));
  vec3 diffuse  = light.diffuse  * diff * vec3(texture(material.diffuse, vTexCoords));
  vec3 specular = light.specular * spec * vec3(texture(material.specular, vTexCoords));
  ambient  *= attenuation;
  diffuse  *= attenuation;
  specular *= attenuation;
  return (ambient + diffuse + specular);
}

// calculates the color when using a spot light.
vec3 CalcSpotLight(SpotLight light, vec3 normal, vec3 fragPos, vec3 viewDir)
{
  vec3 lightDir = normalize(light.position - fragPos);
  
  // diffuse shading
  float diff = max(dot(normal, lightDir), 0.0);
  
  // specular shading
  vec3 reflectDir = reflect(-lightDir, normal);
  float spec = pow(max(dot(viewDir, reflectDir), 0.0), material.shininess);
  
  // attenuation
  float distance = length(light.position - fragPos);
  float attenuation = 1.0 / (light.constant + light.linear * distance + light.quadratic * (distance * distance));    
  
  // spotlight intensity
  float theta = dot(lightDir, normalize(-light.direction)); 
  float epsilon = light.cutOff - light.outerCutOff;
  float intensity = clamp((theta - light.outerCutOff) / epsilon, 0.0, 1.0);
  
  // combine results
  vec3 ambient = light.ambient * vec3(texture(material.diffuse, vTexCoords));
  vec3 diffuse = light.diffuse * diff * vec3(texture(material.diffuse, vTexCoords));
  vec3 specular = light.specular * spec * vec3(texture(material.specular, vTexCoords));
  ambient *= attenuation * intensity;
  diffuse *= attenuation * intensity;
  specular *= attenuation * intensity;
  return (ambient + diffuse + specular);
}